



Anwendung

- Q-Parameter dienen als Platzhalter für die Erstellung variabler Programme

Starres Programm:

L X+20 Y+30 F500

Variables Programm:

Q1=+20

Q2=+30

Q3=+500

L X+Q1 Y+Q2 FQ3





Wo können Q-Parameter verwendet werden?

Variabel sind	Beispiel
Positionen	X+Q1 Y+Q2
Zykluswerte	Q201 = +Q2 ; Tiefe
Schnittdaten	FQ17 SQ10
Kreisdaten	RNDQ12
Werkzeugdaten	TOOL CALL Q1 Z DL+Q2
LBL-Aufruf	CALL LBL Q22 REP Q23 (ab iTNC 30 SW 340422-06)



Arten von Q-Parametern

Art	Bedeutung
Q	Q-Parameter (programmübergreifend)
QL	Lokaler Parameter (kann nur im Programm verwendet werden, nicht programmübergreifend)
QR	Remanenter Parameter , wird über das Ausschalten der Maschine gespeichert
QS	Stringparameter zur Verwendung von Texten



Q-Parameter – Verwendbare Nummern

Bereich	Verwendung
Q1 – Q30	SL-Zyklen (Um eine Überschreibung zu vermeiden, sollte dieser Bereich nicht verwendet werden)
Q31 – Q99	Frei verfügbar
Q100 – Q199	Vorbelegte Parameter mit Sonderfunktion (siehe Benutzerhandbuch)
Q200 – Q1599	Parameter bevorzugt für HEIDENHAIN- und Hersteller-Zyklen (über 400 ab Software iTNC 34049x-xx)
Q1600 – Q1999	Frei verfügbar (ab Software iTNC 34049x-xx)



Parameter – Verwendbare Nummern

Bereich	Verwendung
QL1 – QL499	Frei verfügbar (ab Software 34049x-05)
QR1 – QR499	Frei verfügbar (ab Software 34049x-05)
QS0 – QS1999	Frei verfügbar (ab Software 34049x-03) Ausnahme QS13 (Ausräumwerkzeug) QS18 (Voräum-Werkzeug) QS100 (Werkstoff aus WMAT)




Q-Parameter programmieren

1. Möglichkeit:

- Taste Q im Nummernblock, anschließend über Softkey gewünschte Funktion auswählen, z. B. FN0 „Wert zuweisen“:

FN 0: Q31 =+1

- Dem Q-Parameter Q31 wurde der Wert +1 zugewiesen





Q-Parameter programmieren

2. Möglichkeit:

- Taste Q auf der ASCII-Tastatur (Formeleingabe wird aktiviert)

Q32 =-10

- Dem Q-Parameter Q32 wurde der Wert -10 zugewiesen







Q-Parameter Status lesen

BA Satzfolge/Einzelsatz:

- Programm starten und nach den Q-Parametern mit M0/STOP im Programm stoppen
- Taste Q im Nummernblock
- Übersicht über alle Parameter öffnet
- Q31 und Q32 sind mit Werte belegt

- Über den Softkey  PARAMETER ANZEIGEN können auch die anderen Parameterarten angezeigt werden







Q-Parameter Status lesen

BA Programm Einspeichern:

- Programm starten und nach den Q-Parametern mit M0 M0/STOP im Programm stoppen
- Softkey Q-Info
- Übersicht über alle Parameter öffnet
- Q31 und Q32 sind mit Werte belegt

BA Programm-Test:

- Programm starten und nach den Q-Parametern mit M0 M0/STOP im Programm stoppen
- Taste Q im Nummernblock
- Übersicht über alle Parameter öffnet
- Q31 und Q32 sind mit Werte belegt






Q-Parameter rücksetzen

- Parameter mit 0 befüllen
- Anwahl eines Programmes
- Durch M2, M30 oder END PGM werden die Parameter zurückgesetzt

→ Beachten Sie beim Rücksetzen den Maschinenparameter 7300

```
-----  
;MP7300 Löschen der Status-Anzeige und der Q-Parameter  
;Eingabe: 0 = Status-Anzeige und Q-Parameter und Werkzeug-  
;           Daten löschen, wenn Programm angewählt wird  
;           1 = Status-Anzeige, Q-Parameter und Werkzeug-Daten  
;           löschen bei M02, M30, END PGM und Anwahl  
;           eines Programms  
;           2 = Status-Anzeige und Werkzeug-Daten löschen,  
;           wenn Programm angewählt wird  
;           3 = Status-Anzeige und Werkzeug-Daten löschen,  
;           wenn Programm angewählt wird und bei M02, M30,  
;           END PGM  
;           4 = Status-Anzeige und Q-Parameter löschen,  
;           wenn Programm angewählt wird  
;           5 = Status-Anzeige und Q-Parameter löschen, wenn  
;           Programm angewählt wird und bei M02, M30, END PGM  
;           6 = Status-Anzeige löschen, wenn Programm angewählt  
;           wird  
;           7 = Status-Anzeige löschen, wenn Programm angewählt  
;           wird und bei M02, M30, END PGM  
MP 7300      : 5  
-----
```



Formeleingabe mit Q-Parametern



HEIDENHAIN

iTNC 530

TNC 620

TNC 640



Folgende Rechnung soll im Programm durchgeführt werden:

$10 * 3 + 20$ → das Ergebnis soll im Q50 gespeichert werden

1. Möglichkeit (Eingabe mit Grundfunktionen)

*1 FN 3: Q50 = +10 * +3*

2 FN 1: Q50 = +Q50 + +20

→ Abwärtskompatibles Format bis TNC 155

2. Möglichkeit (Eingabe mit Formel)


*1 Q50 = 10 * 3 + 20*

→ Durch die Formeleingabe können selbst komplexe Rechenoperationen in einem Satz durchgeführt werden



Folgende Funktionen können mit der Formeleingabe verwendet werden:

- Grundfunktionen
- Winkelfunktionen
- Klammern
- Wurzel ziehen
- Quadrieren
- Vor- bzw. Nachkommastellen abschneiden
- Weitere Funktionen sind im Benutzerhandbuch beschrieben



Vorbelegte Q-Parameter



HEIDENHAIN

iTNC 530

TNC 620

TNC 640



Die Steuerung befüllt automatisch bestimmte Parameter im Programm mit Werten, dies sind z.B.:

- Q108 (aktiver Werkzeugradius)
 - Q114 (aktive Werkzeuglänge)
 - Q120 (Berechneter Drehwinkel der A-Achse)
 - Q121 (Berechneter Drehwinkel der B-Achse)
 - Q122 (Berechneter Drehwinkel der C-Achse)
- Die Parameter Q120 – Q122 werden nach einem Schwenkbefehl Zyklus19 bzw. PLANE befüllt

→ Weitere Parameter finden Sie im Benutzerhandbuch



Systemdaten lesen und in Q-Parameter speichern



HEIDENHAIN

iTNC 530

TNC 620

TNC 640



Über die Funktion FN 18 Systemdaten lesen können Informationen aus dem System in einen Q-Parameter gespeichert und weiter verrechnet werden.

Programmierung:

- Taste Q im Nummernblock
- Softkey Sonderfunktion
- Softkey FN18 Lesen SYS-Daten
 - Eingabe des Parameters für das Ergebnis
 - ID (Nummer Systemdatengruppe)
 - NR Nummer Systemdatum
 - IDX Index-Nummer Systemdatum

Die Informationen was bei ID, NR, IDX eingetragen werden muss bekommen Sie aus dem Benutzerhandbuch.

TIPP:

Wenn Sie in der Bearbeitung einer FN18-Funktion sind drücken Sie die HEPL-Taste → Es öffnet sich das Onlinehandbuch und Sie bekommen die gewünschten Informationen.

Programmieren: Q-Parameter / Zusätzliche Funktionen

FN 18: SYS-DATUM READ: Systemdaten lesen

Mit der Funktion FN 18: SYS-DATUM READ können Sie Systemdaten lesen und in Q-Parametern speichern. Die Auswahl des Systemdatums erfolgt über eine Gruppen-Nummer (ID-Nr.), eine Nummer und ggf. über einen Index.


Gruppen-Name, ID-Nr.	Nummer	Index	Bedeutung
Programm-Info, 10	1	-	mm/inch-Zustand
	2	-	Überlappungsfaktor beim Taschenfräsen
	3	-	Nummer aktiver Bearbeitungs-Zyklus
	4	-	Nummer aktiver Bearbeitungs-Zyklus (für Zyklen mit Nummern größer 200)
Maschinenzustand, 20	1	-	Aktive Werkzeug-Nummer
	2	-	Vorbereitete Werkzeug-Nummer
	3	-	Aktive Werkzeug-Achse 0=X, 1=Y, 2=Z, 6=U, 7=V, 8=W
	4	-	Programmierte Spindeldrehzahl
	5	-	Aktiver Spindelzustand: -1=undefiniert, 0=M3 aktiv, 1=M4 aktiv, 2=M5 nach M3, 3=M5 nach M4
	8	-	Kühlmittelzustand: 0=aus, 1=ein
	9	-	Aktiver Vorschub



Beispiel:

Es soll vom aktuellen Werkzeug der Eckenradius R2 aus der TOOL.T gelesen und in Parameter Q60 gespeichert werden:

- 1 *TOOL CALL 5 Z S5000 F1000*
- 2 **FN 18: SYSREAD Q58 = ID20 NR1 ;Lese**
Werkzeugnummer
- 3 **FN 18: SYSREAD Q59 = ID20 NR11 ;Lese**
Werkzeugindex
- 4 **FN 18: SYSREAD Q60 = ID50 NR3 IDXQ58 .Q59**
;Lese R2 aus Zeile Werkzeugnummer (Q58) und
Werkzeugindex (Q59)

T : 5	D10
DOC :	
Z ↓ 	L +60.0000
	R +5.0000
	R2 +3.0000

Q58	=	+5.00000000
Q59	=	+0.00000000
Q60	=	+3.00000000



Bei Q-Parameterprogrammen werden oft mehrere Zustellungen programmiert. Auch diese Zustellungen können über Q-Parameter gesteuert werden.

Hierzu gibt es zwei Varianten:

- CALL LBL 1 REPQ50

- Sprungbefehle

(Q-Taste im Nummernblock / Softkey SPRÜNGE)

- FN9: IF X EQ Y GOTO LBL ...
Wenn Eingabe X gleich Eingabe Y spring zu LBL ...
- FN10: IF X NE Y GOTO LBL ...
Wenn Eingabe X nicht gleich der Eingabe Y spring zu LBL ...
- FN11: IF X GT Y GOTO LBL ...
Wenn Eingabe X größer als Eingabe Y spring zu LBL ...
- FN12: IF X LT Y GOTO LBL ...
Wenn Eingabe X kleiner als Eingabe Y spring zu LBL ...




Programmieren von Programmteiwiederholungen

MW M-TS/ Mai 2013

Beispiel:

Lochreihe mit 8 Bohrungen im Abstand von 10 mm.

Programmiert mit **CALL LBL ... REPQ51**



ITNC530

Programmlauf Satzfolge

Programm-Test

```
0 BEGIN PGM 3 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+50 Z+0
3 ;
4 TOOL CALL "D6" Z S5000 F1000
5 Q50 = 8 ;Anzahl der Bohrungen
6 Q51 = Q50 - 2 ;Wegen
  PGM-Teilwiederholung 1 Bohrung
  weniger, wegen Satz 9 noch eine
  weniger
7 ;
8 CYCL DEF 200 BOHREN
  Q200=+2 ;SICHERHEITS-ABST.
  Q201=-20 ;TIEFE
  Q206= AUTO ;VORSCHUB TIEFENZ.
  Q202=+5 ;ZUSTELL-TIEFE
  Q210=+0 ;VERWEILZEIT OBEN
  Q203=+0 ;KOOR. OBERFLAECHE
  Q204=+50 ;2. SICHERHEITS-ABST.
  Q211=+0 ;VERWEILZEIT UNTEN
9 L X+15 Y+30 R0 FMAX M3 M99
10 LBL 1
11 L IX+10 R0 FMAX M99
12 CALL LBL 1 REPQ51
13 M30
14 END PGM 3 MM
```

4096.00 * T 0:00:41

STOPP BEI START START EINZELS. RESET + START



Programmieren von Programmteiwiederholungen

MW M-TS/ Mai 2013

Beispiel:

Lochreihe mit 4 Bohrungen im Abstand von 10 mm.

Es wird nur der Q50 verändert

ITNC530

Programmlauf
Satzfolge

Programm-Test

```
0 BEGIN PGM 3 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+50 Z+0
3 ;
4 TOOL CALL "D6" Z S5000 F1000
5 Q50 = 4 ;Anzahl der Bohrungen
6 Q51 = Q50 - 2 ;Wegen
  PGM-Teilwiederholung 1 Bohrung
  weniger, wegen Satz 9 noch eine
  weniger
7 ;
8 CYCL DEF 200 BOHREN
  Q200=+2 ;SICHERHEITS-ABST.
  Q201=-20 ;TIEFE
  Q206= AUTO ;VORSCHUB TIEFENZ.
  Q202=+5 ;ZUSTELL-TIEFE
  Q210=+0 ;VERWEILZEIT OBEN
  Q203=+0 ;KOOR. OBERFLAECHE
  Q204=+50 ;2. SICHERHEITS-ABST.
  Q211=+0 ;VERWEILZEIT UNTEN
9 L X+15 Y+30 R0 FMAX M3 M99
10 LBL 1
11 L IX+10 R0 FMAX M99
12 CALL LBL 1 REPO51
13 M30
14 END PGM 3 MM
```

4096.00 * T 0:00:20

STOPP BEI START START EINZELS. RESET + START




Programmieren von Programmteiwiederholungen

MW M-TS/ Mai 2013

Beispiel:

Lochreihe mit 8 Bohrungen im Abstand von 10 mm.

Programmiert mit Sprungbefehlen



ITNC530

Programmlauf
Satzfolge

Programm-Test

```
0 BEGIN PGM 3_1 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+50 Z+0
3 ;
4 TOOL CALL "D6" Z S5000 F1000
5 Q50 = 8 ;Anzahl der Bohrungen
6 Q51 = 2 ;Reset Zaehler
7 ;
8 CYCL DEF 200 BOHREN
  Q200=+2 ;SICHERHEITS-ABST.
  Q201=-20 ;TIEFE
  Q205= AUTO ;VORSCHUB TIEFENZ.
  Q202=+5 ;ZUSTELL-TIEFE
  Q210=+0 ;VERWEILZEIT OBEN
  Q203=+0 ;KOOR. OBERFLAECH
  Q204=+50 ;2. SICHERHEITS-ABST.
  Q211=+0 ;VERWEILZEIT UNTEN
9 L X+15 Y+30 R0 FMAX M3 M99
10 LBL 1
11 L IX+10 R0 FMAX M99
12 Q51 = Q51 + 1 ; Erhoehe um eine Bohrung
13 FN 12: IF +Q51 LT +Q50 GOTO LBL 1
14 FN 9: IF +Q51 EQU +Q50 GOTO LBL 1
15 M30
16 END PGM 3_1 MM
```

0 : 00 : 41

STOPP BEI

START

START EINZELS.

RESET + START



Programmieren von Programmteiwiederholungen

Beispiel:

Lochreihe mit 4 Bohrungen im Abstand von 10 mm.

Es wird nur der Q50 verändert


Programmlauf
Satzfolge

Programm-Test

```
0 BEGIN PGM 3_1 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+50 Z+0
3 ;
4 TOOL CALL "D6" Z S5000 F1000
5 Q50 = 4 ;Anzahl der Bohrungen
6 Q51 = 2 ;Reset Zaehler
7 ;
8 CYCL DEF 200 BOHREN
  Q200=+2 ;SICHERHEITS-ABST.
  Q201=-20 ;TIEFE
  Q205= AUTO ;VORSCHUB TIEFENZ.
  Q202=+5 ;ZUSTELL-TIEFE
  Q210=+0 ;VERWEILZEIT OBEN
  Q203=+0 ;KOOR. OBERFLAECH
  Q204=+50 ;2. SICHERHEITS-ABST.
  Q211=+0 ;VERWEILZEIT UNTEN
9 L X+15 Y+30 R0 FMAX M3 M99
10 LBL 1
11 L IX+10 R0 FMAX M99
12 Q51 = Q51 + 1 ; Erhoehe um eine
  Bohrung
13 FN 12: IF +Q51 LT +Q50 GOTO LBL 1
14 FN 9: IF +Q51 EQU +Q50 GOTO LBL 1
15 M30
16 END PGM 3_1 MM
```

4096.00 * T 0:00:20

STOPP BEI START START EINZELS. RESET + START



Praxisbeispiele



HEIDENHAIN

iTNC 530

TNC 620

TNC 640



Beispiel:

Es sollen die Vorschübe über Parameter variabel gehalten werden:

- 1 $Q50 = 2500$; F XY-Ebene
- 2 $Q51 = 1000$; F Z-Zustellung
- 3 $Q52 = 2000$; F XYZ-Spiralzustellung
- 4 $Q53 = 20000$; F Rückzug

Die Parameter können nun im Bearbeitungsprogramm programmiert werden.

```
0 BEGIN PGM 4 MM
1 Q50 = 2500 ;F XY-Ebene
2 Q51 = 1000 ;F Z-Zustellung
3 Q52 = 2000 ;F XYZ-Spiralbewegung
4 Q53 = 20000 ; Rueckzug
5 ;
6 TOOL CALL "D16" Z S5000
7 M3
8 CYCL DEF 251 RECHTECKTASCHE
  Q215=+0 ;BEARBEITUNGS-UMFANG
  Q218=+80 ;1. SEITEN-LAENGE
  Q219=+50 ;2. SEITEN-LAENGE
  Q220=+0 ;ECKENRADIUS
  Q368=+0 ;AUFMASS SEITE
  Q224=+0 ;DREHLAGE
  Q367=+0 ;TASCHENLAGE
  Q207=+Q50 ;VORSCHUB FRAESEN
  Q351=+1 ;FRAESART
  Q201=-20 ;TIEFE
  Q202=+5 ;ZUSTELL-TIEFE
  Q369=+0 ;AUFMASS TIEFE
  Q206=+Q52 ;VORSCHUB TIEFENZ.
  Q338=+0 ;ZUST. SCHLICHTEN
  Q200=+2 ;SICHERHEITS-ABST.
  Q203=+0 ;KOOR. OBERFLAECHE
  Q204=+50 ;2. SICHERHEITS-ABST.
  Q370=+1 ;BAHN-UEBERLAPPUNG
  Q366=+1 ;EINTAUCHEN
  Q385=+500 ;VORSCHUB SCHLICHTEN
9 L X+0 Y+0 R0 FQ53 M99
10 END PGM 4 MM
```



Beispiel:

Sechseck mit Schlüsselweite 80
Programmiert ohne Q-Parameter

The screenshot shows a CNC control interface with the following components:

- Programmlauf Satzfolge:** A list of G-code commands:

```
0 BEGIN PGM 5 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-20
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 8 Z S5000
4 ;
5 CC X+0 Y+0
6 LP PR+65 PA+0 R0 FMAX M3
7 L Z+10 R0 FMAX
8 L Z-5 R0 F5000
9 LP PR+46.188 PA+0 RL F1000
10 LBL 1
11 LP IPA-60
12 CALL LBL 1 REP 5
13 L Z+100 R0 FMAX M30
14 END PGM 5 MM
```
- 3D Model:** A 3D visualization of a hexagonal part with a central hole and chamfered edges. The model is shown in a perspective view.
- Control Panel:** A set of buttons for controlling the machine, including 'STOPP BEI', 'START', 'START EINZELS.', and 'RESET + START'. There are also buttons for 'XVZ..' with '+' and '-' signs, and a 'WAVE' button labeled 'AUS' and 'EIN'.
- Status Bar:** Displays '31 H +52 V' and '0:00:25'.

Variables Programm:

Was soll variabel werden?

- **Schlüsselweite**
- **Position Mitte X und Y**
- **Startwinkel**



Änderungen im Programm:

- Die Schlüsselweite kann nicht 1:1 im Programm verwendet werden, da das Sechseck mit Polarkoordinaten programmiert wurde.
- Aus der vorgegebenen Schlüsselweite muss der Polarradius berechnet werden.
- Position Mitte in X und Y werden in den CC eingetragen
- Startwinkel wird in PA beim Anfahren eingetragen
- Die Vorpositionierung muss je nach Größe der Schlüsselweite angepasst werden

Änderungen im Programm:

- Die Schlüsselweite kann nicht 1:1 im Programm verwendet werden, da das Sechseck mit Polarkoordinaten programmiert wurde.

Formel für das Programm:

Über die COS-Funktion wird mit Ankathete und Winkel die Hypotenuse berechnet.


Ladewerte:

Q50 = 80 ; Schlüsselweite

Rechnung:

$$QL1 = (Q50 / 2) / \cos 30$$

→ Der Polarradius steht im QL1





Änderungen im Programm:

- Die Vorpositionierung muss je nach Größe der Schlüsselweite angepasst werden

Formel für das Programm:


Schlüsselweite + zweimal der Werkzeugradius

→ Werkzeugradius wird in Q108 automatisch nach dem TOOL CALL hinterlegt

Rechnung:

$$QL2 = QL1 + 2 * Q108$$

→ Die Position für die Vorpositionierung steht im QL2





Beispiel:

Sechseck mit Schlüsselweite 80
Programmiert mit Q-Parameter

ITNC530

Programm-Test

```
0 BEGIN PGM 5_1 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-20
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 8 Z S5000
4 ;
5 * - Ladewerte
6 Q50 = 80 ;Schlüsselweite
7 Q51 = 0 ;Startwinkel
8 Q52 = 0 ;Mitte X
9 Q53 = 0 ;Mitte Y
10 ;
11 * - Berechnungen
12 QL1 = ( Q50 / 2 ) / COS 30 ;PR
    berechnen
13 QL2 = QL1 + 2 * Q108 ;Vorposition
14 ;
15 CC X+Q52 Y+Q53
16 LP PR+QL2 PA+Q51 R0 FMAX M3
17 L Z+10 R0 FMAX
18 L Z-5 R0 F5000
19 LP PR+QL1 PA+Q51 RL F1000
20 LBL 1
21 LP IPA-60
22 CALL LBL 1 REP5 5
23 L Z+100 R0 FMAX M30
24 END PGM 5_1 MM
```

4096.00 * T 0:00:25

STOPP BEI START START EINZELS. RESET + START



Beispiel:

Programmiert mit Q-Parameter

```
0 BEGIN PGM 5_2 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+200 Y+200 Z+0
3 TOOL CALL 8 Z S5000
4 ;
5 * - Ladewerte
6 Q50 = 120 ;Schlüsselweite
7 Q51 = 45 ;Startwinkel
8 Q52 = 100 ;Mitte X
9 Q53 = 80 ;Mitte Y
10 ;
11 * - Berechnungen
12 QL1 = ( Q50 / 2 ) / COS 30 ;PR
    berechnen
13 QL2 = QL1 + 2 * Q108 ;Vorposition
14 ;
15 CC X+Q52 Y+Q53
16 LP PR+QL2 PA+Q51 R0 FMAX M3
17 L Z+10 R0 FMAX
18 L Z-5 R0 F5000
19 LP PR+QL1 PA+Q51 RL F1000
20 LBL 1
21 LP IPA-60
22 CALL LBL 1 REPR 5
23 L Z+100 R0 FMAX M30
24 END PGM 5_2 MM
```

- Schlüsselweite

→ In 120 mm geändert

- Position Mitte X und Y

→ In X 100 mm geändert

→ In Y 80 mm geändert

- Startwinkel

→ In 45° geändert

WICHTIG: Ab Satz 11 wurden keine Veränderungen durchgeführt.



Beispiel:

Zähler mit QR-Parameter

ITNC530

Programmlauf
Satzfolge

Programm-Test

```
0 BEGIN PGM 6 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-20
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 OOL CALL 8 Z S5000
4 QR1 = 0 ;Zaehler RESET
5 ;
6 * - Ladewerte
7 Q50 = 80 ;Schlüsselweite
8 Q51 = 0 ;Startwinkel
9 Q52 = 0 ;Mitte X
10 Q53 = 0 ;Mitte Y
11 ;
12 * - Berechnungen
13 QL1 = ( Q50 / 2 ) / COS 30 ;PR
   berechnen
14 QL2 = QL1 + 2 * Q108 ;Vorposition
15 ;
16 CC X+Q52 Y+Q53
17 LP PR+QL2 PA+Q51 R0 FMAX M3
18 L Z+10 R0 FMAX
19 L Z-5 R0 F5000
20 LP PR+QL1 PA+Q51 RL F1000
21 LBL 1
22 LP IPA-60
23 CALL LBL 1 REPR 5
24 ;
25 ;Bearbeitung beendet Zaehler
   hochzaehlen
26 QR1 = QR1 + 1
27 L Z+100 R0 FMAX M30
28 END PGM 6 MM
```

4096.00 * T 0:00:25

BLK-FORM ANZEIGEN AUSBLEND. ROHTEIL ZURÜCK-SETZEN WERKZEUGE ANZEIGEN AUSBLEND. AUS EIN AUS EIN

Variables Programm:

Was soll variabel werden?

- Nach jedem Werkstück den Zähler um eins erhöhen
- Zähler sollte zurückgesetzt werden können

→ Rücksetzen über Softkey Blockskip



Beispiel:

Gravieren mit QS-Parameter

The screenshot shows the ITNC530 control interface. The top bar displays 'Programm-Test'. The main window is divided into a code editor on the left and a 3D model of a part on the right. The code editor contains the following program:

```
0 BEGIN PGM 6 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-20
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 8 Z S5000
/ 4 QR1 = 0 ;Zaehler RESET
5 ;
6 * - Ladewerte
7 Q50 = 80 ;Schlüsselweite
8 Q51 = 0 ;Startwinkel
9 Q52 = 0 ;Mitte X
10 Q53 = 0 ;Mitte Y
11 ;
12 * - Berechnungen
13 QL1 = ( Q50 / 2 ) / COS 30 ;PR
berechnen
14 QL2 = QL1 + 2 * Q108 ;Vorposition
15 ;
16 CC X+Q52 Y+Q53
17 LP PR+QL2 PA+Q51 R0 FMAX M3
18 L Z+10 R0 FMAX
19 L Z-5 R0 F5000
20 LP PR+QL1 PA+Q51 RL F1000
21 LBL 1
22 LP IPA-60
23 CALL LBL 1 REPQ 5
24 ;
25 ;Bearbeitung beendet Zaehler
hochzaehlen
26 QR1 = QR1 + 1
27 L Z+100 R0 FMAX M30
28 END PGM 6 MM
```

The 3D model shows a blue octagonal part with a grey border. The status bar at the bottom of the window displays '4096.00 * T' and '0:00:25'. The bottom toolbar contains several buttons: 'BLK-FORM ANZEIGEN AUSBLEND.', 'ROHTEIL ZURÜCK-SETZEN', 'WERKZEUGE ANZEIGEN AUSBLEND.', 'AUS EIN', and 'EIN AUS'.

Variables Programm:

Was soll variabel werden?

- Die aktuelle Teilenummer soll auf das Teil graviert werden:

Text: TEIL „aktueller Zählerstand“



Beispiel:

Text als Variable hinterlegen

The screenshot shows the TNC530 control interface. The main window is titled "Programm-Einspeichern/Editieren" and contains the following G-code:

```
3 TOOL CALL 8 Z S5000
/ 4 QR1 = 0 ;Zaehler RESET
5 ;
6 * - Ladewerte
7 Q50 = 80 ;Schlüsselweite
8 Q51 = 0 ;Startwinkel
9 Q52 = 0 ;Mitte X
10 Q53 = 0 ;Mitte Y
11 Q51 = "TEIL"
12 ;
13 * - Berechnungen
14 QL1 = ( Q50 / 2 ) / COS 30 ;PR
berechnen
15 QL2 = QL1 + 2 * Q108 ;Vorposition
16 ;
17 CC X+Q52 Y+Q53
18 LP PR+QL2 PA+Q51 R0 FMAX M3
19 L Z+10 R0 FMAX
20 L Z-5 R0 F5000
21 LP PR+QL1 PA+Q51 RL F1000
22 LBL 1
23 LP IPA-60
24 CALL LBL 1 REP 5
25 ;
26 ;Bearbeitung beendet Zaehler
hochzaehlen
27 QR1 = QR1 + 1
28 L Z+100 R0 FMAX M30
29 END PGM 7 MM
```

The numeric keypad on the right side of the interface has the "Q" key highlighted in blue. The keypad also includes buttons for "XVZ..", "+", "-", "AUS", and "EIN".

Über die Q-Taste im Nummernblock die STRINGFORMEL aufrufen und QS1 = „TEIL“ programmieren



Beispiel:

Text als Variable hinterlegen

The screenshot shows the ITNC530 control interface. The top left corner has 'Manueller Betrieb' and 'Programm-Test'. The main area displays a CNC program with the following code:

```
19 LP PR+QL2 PA+Q51 R0 FMAX M3
20 L Z+10 R0 FMAX
21 L Z-5 R0 F5000
22 LP PR+QL1 PA+Q51 RL F1000
23 LBL 1
24 LP IPA-60
25 CALL LBL 1 REPR 5
26 ;
27 TOOL CALL "STICHEL" Z S1000
28 CYCL DEF 225 GRAVIEREN
  Q550= QS2 ;GRAVIERTEXT
  Q513=+15 ;ZEICHENHOEHE
  Q514=+1 ;FAKTOR ABSTAND
  Q515=+0 ;SCHRIFTART
  Q516=+0 ;TEXTANORDNUNG
  Q374=+0 ;DREHLAGE
  Q517=+50 ;KREISRADIUS
  Q207=+500 ;VORSCHUB FRAESEN
  Q201=-2 ;TIEFE
  Q206=+150 ;VORSCHUB TIEFENZ.
  Q200=+2 ;SICHERHEITS-ABST.
  Q203=+0 ;KOOR. OBERFLAECHE
  Q204=+50 ;2. SICHERHEITS-ABST.
29 L X-25 Y-7.5 R0 FMAX M3 M99
30 ;
31 ;Bearbeitung beendet Zaehler
hochzaehlen
32 QR1 = QR1 + 1
33 L Z+100 R0 FMAX M30
34 END PGM 7 MM
```

The 3D model shows a blue octagonal part with the text 'TEIL 7' engraved in the center. The status bar at the bottom indicates '4096.00 * T' and '0:00:58'. The bottom control panel includes buttons for 'STOPP BEI', 'START', 'START EINZELS.', and 'RESET + START'.

Nun muss der QS1 mit dem Wert aus dem QR1 verbunden werden:

- Stringformel wählen QS2 → Enter
- Ersten Teil des Strings eingeben QS1 → Enter
- Softkey TOCHAR (Umwandeln der Zahl in einen Text)

NC-Satz:

```
16 QS2 = QS1 !! TOCHAR( DAT+QR1 )
```

